

Vacation Hours & Pay Accrual

The following document contains detailed instructions on how to set up a vacation hours and pay accrual. The rate and method of vacation accrual varies between companies. For this example, we use the common method of length of employment to determine the accrual rate. The table below shows the calculation that this sample will illustrate:

<u>Service</u>	<u>Vac/Month</u>	<u>Bi-Monthly</u>	<u>Accrual/Pay Period</u>
3 mths-5 years	6.67 x	12/24	= 3.335
5-10 years	10.00 x	12/24	= 5.000
10+ years	13.33 x	12/24	= 6.665

To properly accrue vacation hours and pay we need to keep track of four different amounts:

- Vacation Hours Due
- Vacation Hours Taken
- Vacation Dollars Due
- Vacation Dollars Taken

To perform the necessary calculations, units due and taken need to be tracked separately. This allows automatic accrual to occur during a pay run when the employee is also being paid for vacation taken.

The hours and dollars accrual cannot be combined in the same formula since we need to increase the dollars paid-to-date while decreasing the accrued vacation units (hours) to date. Within Gold Payroll, the hours and dollars either both go up or both go down with a single pay ID formula. The hours from the vacation pay cannot be used for this since we need to accumulate units life-to-date, but do not need to accumulate the dollars.

SET UP INSTRUCTIONS

Set up the following four formulas. In Payroll, select Tools, Formulas.

Note: The record needed to pull the associated field is denoted in brackets []. For example, the field Hire Date [employee] will be found on the employee record.

Formula Name: LTD Vacation Units

LOOKUP (YTD Units [PR Employee Pay], Employee [PR Check], 5, "V_ACCRU1") –
LOOKUP (YTD Units [PR Employee Pay], Employee
[PR Check], 5, "V_ACCRU2")

This formula retrieves the vacation hours accrued and subtracts the vacation hours taken for this employee. The label of YTD Units is really a life-to-date figure because these accumulators are retained at year end, provided the employee is allowed to carry over unused vacation to the next year. Otherwise, when setting up the Pay ID's simply don't check the Pay ID to "Retain YTD totals at close year". This formula is created for use within the Vac Hours Accrual formula. This formula can also be placed on the Payroll check stub to show employees their total vacation hours available.

Formula Name: Vac Hours Accrual

IF(Check Sequence [PR Check] = "First" AND NEWDATE (Hire Date [PR Employee], 0, 90) < Period End Date [PR Check] AND LTD Vacation Units [Formula] < 999, IF (Period End Date [PR Check] <= NEWDATE (Hire Date [PR Employee], 0, 0,5), MIN ((999 - LTD Vacation Units [Formula]), 3.335), IF (Period End Date [PR Check] <= NEWDATE (Hire Date [PR Employee], 0, 0, 10), MIN ((999 - LTD Vacation Units [Formula]), 5.000), MIN ((999 - LTD Vacation Units [Formula]), 6.665))))

This formula will verify the check sequence to ensure it is the first check of the pay period (a provision for multiple checks). It will then compare the employee's hire date to the period end date of the payroll run to see if the employee's probationary period of 90 days has been completed. The formula then confirms that the employee's net life-to-date hours is less than the stipulated maximum of 999 hours. If any of these conditions are not met, the system will not accrue any vacation hours. If all the conditions are met, the formula will continue to test the employee's hire date to determine which bracket the employee falls into for their vacation accrual rate. This formula also contains a provision to check that this pay period's accrual will not send the net accumulated hours over the 999 hour limit. Due to the complexity of this formula, we have provided a syntax version of the formula below:

- If the check is the employee's first check in this pay period, it has been more than 90 days since this employee was hired and less than 5 years, and they haven't reached their maximum vacation accrual of 999 hours, then...
- Use the minimum of either 3.335 hours or the remaining vacation hours they can accrue before reaching the limit of 999 hours.
- If they were an employee for more than 5 years and less than 10 years, then...
- Use the minimum of either 5.000 hours or the remaining vacation hours they can accrue before reaching the limit of 999 hours.
- Otherwise, we assume they have been employed for over 10 years and are entitled to the 6.665 hours accrued.

Formula Name: Vac Dollars Accrual

LOOKUP (Units [PR Check Pay], Employee [PR Check], Period End Date [PR Check], Check Sequence [PR Check], "V_ACCRU1") * LOOKUP (Amount [PR Employee Pay], Employee [PR Check], 1, "REG")

This formula multiplies the number of hours the Vac Hours Accrual formula calculated by the hourly rate stored on the employee record for regular pay. This formula assumes that the employee's regular Pay ID is named REG.

Formula Name: Vac Dollars Paid

IF(LOOKUP(Units [PR Check Pay], Employee [PR Check], Period End Date [PR Check], Check Sequence [PR Check], "V_ACCRU2") > 0, LOOKUP (Units [PR Check Pay], Employee [PR Check], Period End Date [PR Check], Check Sequence [PR Check], "V_ACCRU2") * LOOKUP (Amount [PR Employee Pay], Employee [PR Check], 1, "REG"))

This formula looks to see if any vacation hours were taken this pay period. If yes, the vacation hours are multiplied by the employee's regular pay rate. This allows an increase in the vacation hours taken and corresponding vacation pay. The formula assumes that the employee's regular Pay ID is named REG.

Set Up the Pay ID's

You will need to set up each of the following Pay ID's. In Payroll, from the Setup menu, select Pays.

Pay ID	Desc.	Pay Type	Retain YTD Totals	GL Accounts	Distribute To	Calculation Method	Formula Name/Amount	Employee Pay Record
V_ACCRU1	Vacation Hrs Accrued	Accrual (units)	Yes, unless employees are not allowed to carry over vacation	No	No distribution unless keeping track of units in GL	Formula	Vac Hours Accrual	Auto Pay
V_DOLR1	Vacation Dollars Accrued	Accrual	No	Appropriate expense account -- Appropriate liability account	Distribute to GL. If appropriate also distribute to JC and/or EQ	Formula	Vac Dollars Accrual	Auto Pay
V_ACCRU2	Vacation Hrs Taken	Accrual (units)	Yes, unless employees are not allowed to carry over vacation	No	No distribution unless keeping track of units in GL	Flat Amount	none	Direct entry in Enter Time
V_DOLR2	Vacation Dollars Paid	Other	No	Enter the vacation accrual liability account	GL	Formula	Vac Dollars Paid	Auto Pay

Set Up the Pay ID's on the Employee Record

Using Setup, Employees, access the employee record that you will be setting up this accrual on. After entering the employee number and returning through their name, select [PAYS]. Enter V_ACCRU1, V_DOLR1, and V_DOLR2 Pay ID's and check the Auto box with a calculation frequency of Check.

The best way to test a Pay ID is in Enter Checks. Enter the Pay ID V_ACCRU2 and the number of vacation hours in the units field to pay the employee. Then use the Process push button to calculate the vacation accrual Pays. The system should calculate an additional hours accrual. It should also pay the employee their vacation dollars according to the vacation hours entered.

Note: The above instructions can also be used to set up a Sick Pay Accrual. You will need to set up four new Pay ID's that correspond with the sick pay. The formulas will need to be modified to include the new Pay ID's for sick pay.